Mat 1. 2-timersprøve den 8. december 2019.

JE 1.12.19

Opgave 1

```
> restart;with(LinearAlgebra):
```

Et homogent lineært ligningssystem er givet ved

$$x_1 + x_2 + x_3 + x_4 = 0$$
$$2\,x_1 - x_2 + 8\,x_3 - 4\,x_4 = 0$$
$$x_1 - 2\,x_2 + 7\,x_3 + a x_4 = 0$$

hvor a er et vilkårligt reelt tal.
Koefficientmatricer er

```
> A:=a-><1,1,1,1;2,-1,8,-4;1,-2,7,a>:
> A(a);
```

$$\begin{bmatrix} 1 & 1 & 1 & 1 \\ 2 & -1 & 8 & -4 \\ 1 & -2 & 7 & a \end{bmatrix} \qquad \textbf{(1.1)}$$

Spørgsmål 1

For $a = 1$ er totalmatricen

```
> T:=<A(1)|0,0,0>;
```

$$T := \begin{bmatrix} 1 & 1 & 1 & 1 & 0 \\ 2 & -1 & 8 & -4 & 0 \\ 1 & -2 & 7 & 1 & 0 \end{bmatrix} \qquad \textbf{(1.1.1)}$$

som har trappeformen

```
> trap('T'):=ReducedRowEchelonForm(T);
```

$$trap(T) := \begin{bmatrix} 1 & 0 & 3 & 0 & 0 \\ 0 & 1 & -2 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix} \qquad \textbf{(1.1.2)}$$

Det fuldstændigt reducerede lineære ligningssystem er da

$x_1 + 3\,x_3 = 0$

$x_2 - 2\,x_3 = 0$

$x_4 = 0$

Sættes $x_3 = t$ fås den fuldstændige løsning på standardparameterform

$(x_1, x_2, x_3, x_4) = t\,(-3, 2, 1, 0)$, $t \in \mathbb{R}$.

Spørgsmål 2

```
> T:=<A(a)|0,0,0>;
```

$$T := \begin{bmatrix} 1 & 1 & 1 & 1 & 0 \\ 2 & -1 & 8 & -4 & 0 \\ 1 & -2 & 7 & a & 0 \end{bmatrix} \quad \textbf{(1.2.1)}$$

```
> T2:=RowOperation(T,[2,1],-2);
```

$$T2 := \begin{bmatrix} 1 & 1 & 1 & 1 & 0 \\ 0 & -3 & 6 & -6 & 0 \\ 1 & -2 & 7 & a & 0 \end{bmatrix} \quad \textbf{(1.2.2)}$$

```
> T3:=RowOperation(T2,[3,1],-1);
```

$$T3 := \begin{bmatrix} 1 & 1 & 1 & 1 & 0 \\ 0 & -3 & 6 & -6 & 0 \\ 0 & -3 & 6 & a-1 & 0 \end{bmatrix} \quad \textbf{(1.2.3)}$$

Af **T**3 aflæses, at $\rho(\mathbf{A}(a)) = 2 \Leftrightarrow a-1=-6 \Leftrightarrow a=-5$.
For $a=-5$ er totalmatricen

```
> T:=<A(-5)|0,0,0>;
```

$$T := \begin{bmatrix} 1 & 1 & 1 & 1 & 0 \\ 2 & -1 & 8 & -4 & 0 \\ 1 & -2 & 7 & -5 & 0 \end{bmatrix} \quad \textbf{(1.2.4)}$$

som har trappeformen

```
> trap('T'):=ReducedRowEchelonForm(T);
```

$$trap(T) := \begin{bmatrix} 1 & 0 & 3 & -1 & 0 \\ 0 & 1 & -2 & 2 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix} \quad \textbf{(1.2.5)}$$

Det fuldstændigt reducerede lineære ligningssystem er da
$x_1 + 3\,x_3 - x_4 = 0$
$x_2 - 2\,x_3 + 2\,x_4 = 0$
Sættes $x_3 = t_1$ og $x_4 = t_2$ fås den fuldstændige løsning på standardparameterform
$(x_1, x_2, x_3, x_4) = t_1(-3, 2, 1, 0) + t_2(1, -2, 0, 1)\ \ ,\ t_1, t_2 \in \mathbb{R}$.
For $(t_1, t_2) = (1, 0)$ fås løsningen $(-3, 2, 1, 0)$ og for $(t_1, t_2) = (0, 1)$ fås løsningen $(1, -2, 0, 1)$.
Disse to løsninger er lineært uafhængige, da de ikke er proportionale.
For $a=-5$ er $(-3, 2, 1, 0)$ og $(1, -2, 0, 1)$ således to lineært uafhængige løsninger til ligningssystemet.

Opgave 2

```
> restart;with(LinearAlgebra):
```

Et 2-dimensionalt vektorrum V har basen $v = (\mathbf{v}_1, \mathbf{v}_2)$.

$f: \mathrm{V} \to \mathbb{R}^3$ er lineær og afbildningsmatricen ${}_e\mathbf{F}_v = [\,{}_e f(\mathbf{v}_1)\;{}_e f(\mathbf{v}_2)]$ er

```
> eFv:=<1,2;0,-1;2,0>;
```

$$eFv := \begin{bmatrix} 1 & 2 \\ 0 & -1 \\ 2 & 0 \end{bmatrix} \tag{2.1}$$

Spørgsmål 1

Vektoren $\mathbf{v}_3 = 2\,\mathbf{v}_1 - 5\,\mathbf{v}_2$ har koordinatvektoren ${}_v\mathbf{v}_3 = \begin{bmatrix} 2 \\ -5 \end{bmatrix}$ med hensyn til basen v.

${}_e f(\mathbf{v}_3) = {}_e\mathbf{F}_v\;{}_v\mathbf{v}_3$

```
> eFv.<2,-5>;
```

$$\begin{bmatrix} -8 \\ 5 \\ 4 \end{bmatrix} \tag{2.1.1}$$

Dvs $f(\mathbf{v}_3) = (-8, 5, 4)$.

Spørgsmål 2

$f(\mathbf{v}) = (1, 2, 10) \Leftrightarrow {}_e f(\mathbf{v}) = \begin{bmatrix} 1 \\ 2 \\ 10 \end{bmatrix} \Leftrightarrow {}_e\mathbf{F}_v\;{}_v\mathbf{v} = \begin{bmatrix} 1 \\ 2 \\ 10 \end{bmatrix}$.

Totalmatricen er

```
> T:=<eFv|1,2,10>;
```

$$T := \begin{bmatrix} 1 & 2 & 1 \\ 0 & -1 & 2 \\ 2 & 0 & 10 \end{bmatrix} \tag{2.2.1}$$

som har trappeformen

```
> trap('T'):=ReducedRowEchelonForm(T);
```

$$trap(T) := \begin{bmatrix} 1 & 0 & 5 \\ 0 & 1 & -2 \\ 0 & 0 & 0 \end{bmatrix} \tag{2.2.2}$$

Heraf aflæses, at ${}_v\mathbf{v} = \begin{bmatrix} 5 \\ -2 \end{bmatrix}$ og dermed er løsningsvektoren $\mathbf{v} = 5\,\mathbf{v}_1 - 2\,\mathbf{v}_2$.

Spørgsmål 3

$f(\mathrm{V}) = \mathrm{span}\{f(\mathbf{v}_1), f(\mathbf{v}_2)\} = \mathrm{span}\{(1, 0, 2), (2, -1, 0)\}$.
$\dim(f(\mathrm{V})) = \rho({}_e\mathbf{F}_v) = 2$. (Følger feks af trap(**T**) i spørgsmål 2.)
Billedrummet $f(\mathrm{V}) = \mathrm{span}\{(1, 0, 2), (2, -1, 0)\}$ er således en plan i $\mathbb{R}^3$ gennem origo.

$\dim(\ker(f)) = \dim(\mathrm{V}) - \dim(f(\mathrm{V})) = 0$.

Spørgsmål 4

$\mathbf{u} = (u_1, u_2, u_3) \in f(\mathrm{V}) \Leftrightarrow f(\mathbf{x}) = \mathbf{u}$ har en løsning $\mathbf{x} = x_1\mathbf{v}_1 + x_2\mathbf{v}_2 \in \mathrm{V} \Leftrightarrow {}_e\mathbf{F}_v \begin{bmatrix} x_1 \\ x_2 \end{bmatrix} = \begin{bmatrix} u_1 \\ u_2 \\ u_3 \end{bmatrix}$ har en løsning $(x_1, x_2) \in \mathbb{R}^2$.
Hvis $\mathbf{u} = \mathbf{e}_1 = (1, 0, 0)$ så har matrixligningen totalmatricen

```
> T:=<eFv|1,0,0>;
```

$$T := \begin{bmatrix} 1 & 2 & 1 \\ 0 & -1 & 0 \\ 2 & 0 & 0 \end{bmatrix} \quad \textbf{(2.4.1)}$$

som har trappeformen

```
> trap('T'):=ReducedRowEchelonForm(T);
```

$$trap(T) := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad \textbf{(2.4.2)}$$

Da $\rho({}_e\mathbf{F}_v) = 2 < 3 = \rho(\mathbf{T})$ så har matrixligningen ingen løsninger hvilket betyder, at $\mathbf{e}_1 = (1, 0, 0)$ ikke tilhører billedrummet $f(\mathrm{V})$. (Vektoren $(1, 0, 0)$ ligger altså ikke i planen $f(\mathrm{V})$ gennem origo udspændt af vektorerne $(1, 0, 2)$ og $(2, -1, 0)$.)

Opgave 3

```
> restart;with(LinearAlgebra):
> prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
```

$\mathbb{R}^3$ udstyret med det sædvanlige prikprodukt.

```
> v1:=<1,-1,1>;
```

$$v1 := \begin{bmatrix} 1 \\ -1 \\ 1 \end{bmatrix} \quad \textbf{(3.1)}$$

```
> v2:=<1,0,-1>;
```

$$v2 := \begin{bmatrix} 1 \\ 0 \\ -1 \end{bmatrix} \tag{3.2}$$

```
> v3:=<1,1,0>;
```

$$v3 := \begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix} \tag{3.3}$$

Om en reel 3×3-matrix **A** oplyses, at den har egenvektorrummene
$E_6 = \mathrm{span}\{\mathbf{v}_1\}$ og $E_{-3} = \mathrm{span}\{\mathbf{v}_2, \mathbf{v}_3\}$.

Spørgsmål 1

```
> V:=<v1|v2|v3>;
```

$$V := \begin{bmatrix} 1 & 1 & 1 \\ -1 & 0 & 1 \\ 1 & -1 & 0 \end{bmatrix} \tag{3.1.1}$$

```
> Determinant(V);
```

$$3 \tag{3.1.2}$$

Søjlerne i **V** er således tre lineært uafhængige egenvektorer for **A**.
Sættes

```
> Lambda:=DiagonalMatrix([6,-3,-3]);
```

$$\Lambda := \begin{bmatrix} 6 & 0 & 0 \\ 0 & -3 & 0 \\ 0 & 0 & -3 \end{bmatrix} \tag{3.1.3}$$

så er $\boldsymbol{\Lambda} = \mathbf{V}^{-1}\,\mathbf{A}\,\mathbf{V}$.

Spørgsmål 2

$t_1\mathbf{v}_1$, hvor $t_1 \in \mathbb{R}$, er en vilkårlig vektor i E_6 og $t_2\mathbf{v}_2 + t_3\mathbf{v}_3$, hvor $(t_2, t_3) \in \mathbb{R}^2$, er en vilkårlig vektor i E_{-3}.
Da

```
> simplify(prik(t1*v1,t2*v2+t3*v3));
```

$$0 \tag{3.2.1}$$

så er enhver vektor i E_6 ortogonal på enhver vektor i E_{-3}. Dvs de to egenvektorrum E_6 og E_{-3} er ortogonale.

Spørgsmål 3

```
> q1:=v1/sqrt(prik(v1,v1));
```

$$q1 := \begin{bmatrix} \frac{\sqrt{3}}{3} \\ -\frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \end{bmatrix} \quad (3.3.1)$$

er en ortonormal basis for E_6.

```
> q2:=v2/sqrt(prik(v2,v2));
```

$$q2 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ 0 \\ -\frac{\sqrt{2}}{2} \end{bmatrix} \quad (3.3.2)$$

og

```
> q3:=kryds(q1,q2);
```

$$q3 := \begin{bmatrix} \frac{\sqrt{3}\,\sqrt{2}}{6} \\ \frac{\sqrt{3}\,\sqrt{2}}{3} \\ \frac{\sqrt{3}\,\sqrt{2}}{6} \end{bmatrix} \quad (3.3.3)$$

udgør en ortonormal basis for E_{-3}.

(**q**1, **q**2, **q**3) er da en ortonormal basis for $\mathbb{R}^3$ udstyret med det sædvanlige prikprodukt bestående af egenvektorer for **A**.

Sættes

```
> Q:=<q1|q2|q3>;
```

$$Q := \begin{bmatrix} \frac{\sqrt{3}}{3} & \frac{\sqrt{2}}{2} & \frac{\sqrt{3}\,\sqrt{2}}{6} \\ -\frac{\sqrt{3}}{3} & 0 & \frac{\sqrt{3}\,\sqrt{2}}{3} \\ \frac{\sqrt{3}}{3} & -\frac{\sqrt{2}}{2} & \frac{\sqrt{3}\,\sqrt{2}}{6} \end{bmatrix} \quad (3.3.4)$$

så er **Q** positiv ortogonal

```
> Transpose(Q).Q;Determinant(Q);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$1 \tag{3.3.5}$$

og

$$\mathbf{Q}^{-1}\,\mathbf{A}\,\mathbf{Q} = \mathbf{\Lambda} \Leftrightarrow \mathbf{A} = \mathbf{Q}\;\mathbf{\Lambda}\,\mathbf{Q}^{-1} = \mathbf{Q}\,\mathbf{\Lambda}\,\mathbf{Q}^{\mathrm{T}},$$

hvor $\mathbf{\Lambda}$ er den i spørgsmål 1 angivne diagonalmatrix.
Den ukendte matrix er da

```
> A:=Q.Lambda.Transpose(Q);
```

$$A := \begin{bmatrix} 0 & -3 & 3 \\ -3 & 0 & -3 \\ 3 & -3 & 0 \end{bmatrix} \tag{3.3.6}$$

Opgave 4

```
> restart;with(LinearAlgebra):with(plots):
```

$$\begin{bmatrix} x_1'(t) \\ x_2'(t) \end{bmatrix} = \mathbf{A} \begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix}, \; t \in \mathbb{R}$$

hvor $\mathbf{A}$ er en reel 2×2-matrix.
Det oplyses, at systemets fuldstændige komplekse løsning er givet ved.

$$\begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix} = c1\,\mathrm{e}^{(2+6\,i)t} \begin{bmatrix} i \\ 1 \end{bmatrix} + c2\,\mathrm{e}^{(2-6\,i)t} \begin{bmatrix} -i \\ 1 \end{bmatrix}, \mathrm{t} \in \mathbb{R} \;,\; c1, c2 \in \mathbb{C}.$$

Skrevet ud i Maple

```
> x:=unapply(c1*exp((2+6*I)*t)*<I,1>+c2*exp((2-6*I)*t)*<-I,1>,t):
> x(t);
```

$$\begin{bmatrix} \mathrm{I}\,c1\,\mathrm{e}^{(2+6\,\mathrm{I})\,t} - \mathrm{I}\,c2\,\mathrm{e}^{(2-6\,\mathrm{I})\,t} \\ c1\,\mathrm{e}^{(2+6\,\mathrm{I})\,t} + c2\,\mathrm{e}^{(2-6\,\mathrm{I})\,t} \end{bmatrix} \tag{4.1}$$

Spørgsmål 1

Af den givne fuldstændige komplekse løsning aflæses, at samtlige egenværdier for systemmatricen $\mathbf{A}$ er $2 + 6\,i$ og $2 - 6\,i$ med de tilhørende egenvektorrum

$$\mathrm{E}_{2+6\,i} = \mathrm{span}\left\{\begin{bmatrix} i \\ 1 \end{bmatrix}\right\} \text{ og } \mathrm{E}_{2-6\,i} = \mathrm{span}\left\{\begin{bmatrix} -i \\ 1 \end{bmatrix}\right\}.$$

Spørgsmål 2

Den partikulære løsning, hvor

```
> c1:=1/2;
```

$$c1 := \frac{1}{2} \tag{4.2.1}$$

og

```
> c2:=1/2;
```

$$c2 := \frac{1}{2} \tag{4.2.2}$$

er

```
> x(t);
```

$$\left[\begin{array}{c} \frac{\mathrm{I}\,\mathrm{e}^{(2+6\,\mathrm{I})\,t}}{2} - \frac{\mathrm{I}\,\mathrm{e}^{(2-6\,\mathrm{I})\,t}}{2} \\ \frac{\mathrm{e}^{(2+6\,\mathrm{I})\,t}}{2} + \frac{\mathrm{e}^{(2-6\,\mathrm{I})\,t}}{2} \end{array}\right] \tag{4.2.3}$$

Da

```
> 'x(0)'=x(0);
```

$$x(0) = \left[\begin{array}{c} 0 \\ 1 \end{array}\right] \tag{4.2.4}$$

så tilfredsstiller denne partikulære løsning den opstillede begyndelsesbetingelse.

Spørgsmål 3

Den partikulære løsning fra forrige spørgsmål udtrykt i reelle funktioner er

```
> xr:=t->evalc(x(t)):
> xr(t);
```

$$\left[\begin{array}{c} -\mathrm{e}^{2\,t}\sin(6\,t) \\ \mathrm{e}^{2\,t}\cos(6\,t) \end{array}\right] \tag{4.3.1}$$

Den banekurve, som punktet $(x_1(t), x_2(t))$ gennemløber i tiden $t \in [0, 1]$ er

```
> P1:=plot([xr(t)[1],xr(t)[2],t=0..1],scaling=constrained,
  tickmarks=[6,11],color=blue):
> tekst:=textplot([[0.4,1,"P"],[-1.4,-0.3,"Q"],[0.4,-2.55,"R"]]
  ):
> display(P1,tekst);
```



Da $x_2(t) = e^{2t}\cos(6t) = 0 \Leftrightarrow 6t = \frac{\pi}{2} + p\,\pi, p \in \mathbb{Z} \Leftrightarrow t = \frac{\pi}{12} + p\,\frac{\pi}{6}, p \in \mathbb{Z}$, så gennemskæres

førsteaksen første gang for $t = \frac{\pi}{12}$.

Punktet Q på førsteaksen, som passeres for $t = \frac{\pi}{12}$ er

```
> evalf(xr(Pi/12));
```

$$\begin{bmatrix} -1.688091795 \\ 0. \end{bmatrix} \tag{4.3.2}$$

hvilket ser fornuftigt ud i forhold til figuren.

Da $x_1(t) = -e^{2t}\sin(6t) = 0 \Leftrightarrow 6t = p\pi, p \in \mathbb{Z} \Leftrightarrow t = p\,\frac{\pi}{6}, p \in \mathbb{Z}$, så gennemskæres andenaksen

første gang for $t = \frac{\pi}{6}$.

Punktet R på andenaksen, som passeres for $t = \frac{\pi}{6}$ er

```
> evalf(xr(Pi/6));
```

$$\begin{bmatrix} 0. \\ -2.849653908 \end{bmatrix} \tag{4.3.3}$$

hvilket ser fornuftigt ud i forhold til figuren.

Man kunne også direkte benytte den med komplekse tal udtrykte løsning

```
> P2:=plot([x(t)[1],x(t)[2],t=0..1],scaling=constrained,
  tickmarks=[6,11],color=blue):
> tekst:=textplot([[0.4,1,"P"],[-1.4,-0.3,"Q"],[0.4,-2.55,"R"]]
  ):
> display(P2,tekst);
```



Skæring med første aksen

```
> solve(x(t)[2]=0,t,allsolutions);
```

$$-\frac{1}{6}\pi\_Z2\sim -\frac{1}{12}\pi \tag{4.3.4}$$

Første positive løsning er igen $t = \frac{\pi}{12}$.

Skæring med andenaksen

```
> solve(x(t)[1]=0,t,allsolutions);
```

$$-\frac{\pi\ \_Z3\sim}{6} \tag{4.3.5}$$

Første positive løsning er igen $t = \frac{\pi}{6}$.